Graphics Commands PERIODOGRAM

## PERIODOGRAM

#### **PURPOSE**

Generates an auto-periodogram.

#### **DESCRIPTION**

A periodogram is a graphical data analysis technique for examining frequency-domain models of an equi-spaced time series. The periodogram is the Fourier transform of the autocovariance function. An equi-spaced time series is one in which the distance between adjacent points is constant. The periodogram (or spectrum) for a time series  $x_t$  is:

$$S(f) = \frac{\Delta}{n} \left( \left( \sum_{t=-n}^{n-1} x_t \cos(2\pi f t \Delta) \right)^2 + \left( \sum_{t=-n}^{n-1} x_t \sin(2\pi f t \Delta) \right)^2 \right)$$
 (EQ 2-18)

where f is the frequency, n is the number of observations in the time series,  $\Delta$  is (n+1)/2 for n odd and (n+2)/2 for n even. The periodogram then consists of:

Vertical axis = the spectrum estimate at the given frequency;

Horizontal axis = Fourier frequencies (1/n, 2/n, 3/n, ..., (n/2)/n) where n is the number of observations in the time series.

The frequency is measured in cycles per unit time where unit time is defined to be the distance between adjacent points. A frequency of 0 corresponds to an infinite cycle while a frequency of 0.5 corresponds to a cycle of 2 data points. Equi-spaced time series are inherently limited to detecting frequencies between 0 and 0.5.

From a data analysis point of view, the type of structure in the autocovariance function indicates the location of peaks in the periodogram. The variance at a given frequency is also referred to as the average power. Peaks in the periodogram also indicate the dominant frequency for underlying cyclic models. Smooth time series tend to generate periodograms with most of their variance (or power) in the low frequencies while rapidly oscillating series tend to generate periodograms with most of their variance in the higher frequencies. Once the dominant peaks have been identified, the next step is typically to use other time series analysis techniques (such as the complex demodulation phase plots) to determine if this frequency is constant over the entire domain of the data, or to carry out a a nonlinear fit with an underlying cyclic model (see the documentation for the COMPLEX DEMODULATION PLOT). As a simple example of fitting a nonlinear cyclic model, a time series can be modeled as:

$$x_t = \mu + R \times \cos(\omega t + \phi) + \varepsilon_t$$
 (EQ 2-19)

where  $\mu$  is the mean of the time series, R is the amplitude,  $\phi$  is the phase shift,  $\omega$  is the frequency (which can be estimated with the periodogram), and  $\varepsilon_t$  is the residual.

### **SYNTAX**

PERIODOGRAM <y>

<SUBSET/EXCEPT/FOR qualification>

where <y> is a response variable;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

## **EXAMPLES**

PERIODOGRAM Y

PERIODOGRAM Y2 SUBSET Y2 > 2

#### NOTE 1

The spectral plot is a refinement of the periodogram (it smooths the spectrum estimate) and is generally recommended in its place since it has better statistical properties. See the SPECTRAL PLOT command for details.

#### NOTE 2

Missing values are not allowed. It is also common to remove trends by differencing  $(x_t - x_{t-1})$  or to apply some other type of filter before generating the periodogram.

### NOTE 3

Different time series texts present the equation for the periodogram in slightly different forms. However, these forms should be mathematically equivalent. DATAPLOT uses the definition from the Jenkins and Watts book (see below). The Bloomfield text (see below) uses a slightly different, although mathematically equivalent, definition.

PERIODOGRAM Graphics Commands

#### NOTE 4

The spectral density is the spectrum divided by the variance of the times series. The spectral density is the Fourier transform of the autocorrelation function rather than the autocovariance function. This form of the periodogram allows comparisons of the spectrums of times series that may have different scales. The program example below shows how to plot this form of the periodogram.

#### NOTE 5

The CROSS-PERIODOGRAM, CO-PERIODOGRAM, and QUADRATURE PERIODOGRAM commands are recognized. However, these variations are not implemented at this time. These commands will generate a page erase, but no plot is generated. These options are recognized for the SPECTRUM command.

#### NOTE 6

The appearance of the periodogram can be controlled with the proper settings of the LINE, CHARACTER, SPIKES, and BAR commands. The most typical choices are as a connected line segment or as spikes. Some analyst prefer to turn the log scale on for the vertical axis (YLOG ON). If you want log scales, there is typically a problem with zero values. This can be circumvented with the following commands:

```
PERIODOGRAM Y
LET YJUNK = YPLOT
LET YJUNK = 0.000001 SUBSET YPLOT <= 0
LET XJUNK = XPLOT
YLOG ON
PLOT YJUNK XJUNK
```

## **DEFAULT**

None

#### **SYNONYMS**

None

### **RELATED COMMANDS**

SPECTRUM = Generates a spectral plot.

CORRELATION PLOT = Generates a correlation plot.

COMPLEX DEMOD PLOT = Generates a complex demodulation plot.

LAG PLOT = Generates a lag plot.

PLOT = Generates a data or function plot.
4-PLOT = Generates 4-plot univariate analysis.
CHARACTERS = Sets the types for plot characters.
LINES = Sets the types for plot lines.

SPIKES = Sets the on/off switches for plot spikes.

SUMMARY = Generates a table of summary statistics.

LET = Generates sine/cosine transformations (plus much more).

FIT = Carries out a least squares fit.

#### REFERENCE

"Spectral Analysis and Its Applications," Jenkins and Watts, Holden-Day, 1968 (chapters 6 and 7, page 211 for the equation for the spectrum).

"Fourier Analysis of Time Series: An Introduction," Bloomfield, Wiley and Sons, 1976 (section 2.3 for the periodogram).

## **APPLICATIONS**

Frequency Time Series Analysis

# IMPLEMENTATION DATE

Pre-1987

Graphics Commands PERIODOGRAM

### **PROGRAM**

. THIS SAMPLE PROGRAM READS THE FILE LEW.DAT IN THE

. DATAPLOT REFERENCE DIRECTORY. THESE DATA ARE

. BEAM DEFLECTION DATA.

SKIP 25

READ LEW.DAT DEFLECT

.

TITLE AUTOMATIC

MULTIPLOT 2 2; MULTIPLOT CORNER COORDINATES 0 0 100 100

LET A = MEAN DEFLECT

LINE BLANK; SPIKE ON; SPIKE BASE A

PLOT DEFLECT

.

LINE SOLID; SPIKE OFF

YLIMITS 0 40000; YTIC OFFSET 0 2000 Y1LABEL POWER; X1LABEL FREQUENCY

PERIODOGRAM DEFLECT

SPIKE ON; LINE BLANK PERIODOGRAM DEFLECT

.

LET A = VARIANCE DEFLECT; LET TEMP = YPLOT/A; LET X = XPLOT

YLIMITS; YTIC OFFSET 0 0 TITLE SPECTRAL DENSITY PLOT

SPIKE BASE 0 PLOT TEMP X

.

END OF MULTIPLOT

